

Attack of the Slime Creatures

Hold off the Slime Creatures! How long can you blast the terrors as they come closer and closer?

The Slime Creatures appear and disappear as they swim underwater toward Jacque. Then suddenly they appear for a moment as they surface to attack Jacque.

Shoot at them with Jacque's gun by pressing the space bar, and score more points the closer they are. You'll hear a sound if they're hit.

Otherwise, the Positron Blast will have missed because they have dived into the swamp before being hit. But watch out. If you miss, the Swamp Creatures may randomly move up to twice as far compared to when you don't fire. Your only clue as to how long they'll appear is a series of beeps before they surface as their air bubbles rise to the surface. The number of beeps is proportional to how long they'll be visible. Also, as they come closer, the pitch and intensity of

the sounds increase. So, if you listen carefully, you will have important clues as to where the Swamp Creatures appear and for how long.

Game Program Listing

```

2 PRINT"(SC)":PRINT"  ATTACK OF THE":
  PRINT:PRINT"  SLIME CREATURES"
3 PRINT:PRINT:PRINT"PRESS SPACE BAR TO
  FIRE":PRINT:PRINT:PRINT
5 POKE808,114
10 POKE52,28:POKE56,28:CLR
20 CS=256*PEEK(52)+PEEK(51)
30 FORI=CS TOCS+511:POKEI,
  PEEK(I+32768-CS):NEXT
32 FORK=1TO12:READX:POKE7511+K,X:NEXT
34 DATA169,239,141,32,145,173,33,145,141,
  101,29,96
40 FORI=0TO7:READJ:POKECS+37*8+I,J:NEXT
50 DATA24,152,255,61,60,60,228,4
52 FORI=0TO7:READJ:POKECS+27*8+I,J:NEXT
53 DATA255,247,247,119,182,213,251,7
60 FORI=0TO7:READJ:POKECS+30*8+I,J:NEXT
70 DATA227,201,145,1,113,241,161,129
72 FORI=0TO7:READJ:POKECS+29*8+I,J:NEXT
73 DATA240,239,213,182,119,247,247,255
80 FORI=0TO7:READJ:POKECS+31*8+I,J:NEXT
90 DATA193,225,241,240,240,240,240,248
100 FORI=0TO7:READJ:POKECS+28*8+I,J:NEXT
110 DATA255,255,0,0,255,255,255,255

```



```

114 FORI=0TO7:READJ:POKECS+I,J:NEXT
116 DATA255,255,255,255,255,255,255,255
120 POKE36869,255:PRINT"{SC}"
121 PRINT"NAME? (19 LETTERS MAX)":
    PRINT:INPUTN$:PRINT"{HM}":PRINT"{SC}"
122 FORI=7680TO7745:POKEI,0:POKEI+30720,5:
    NEXT
125 X$="SCORE = ":PRINTSPC(5)X$:
    PRINT"{HM}"
130 I=3:S=7680:CL=38400:DI=30720
140 SC=0:X=0
145 IFTA=2THEN440
290 FORJJ=22TO11STEP-1
300 FORII=IT021
310 POKES+22*JJ+II,0:POKECL+22*JJ+II,2
320 NEXTII
330 X=X+1
340 IFX>3THEN350
345 GOTO370
350 I=I+1
360 X=0
370 NEXTJJ
430 REM **INITIAL POSITIONS
440 POKES+22*12+4,37:POKECL+22*12+4,6
450 M=INT(9*RND(X))+10
460 M=M-INT(3*RND(X))-1
462 FORJO=1TO100*M
463 NEXTJO

```

```

465 IFM<6THEN730
470 GOSUB900
472 TT=RND(X)*M
473 FORJJ=1TOTT
475 POKE36878,17-M/2:POKE36875,279-M*8:
    FORT=1TO50:NEXT:POKE36875,0
477 NEXT
490 GOSUB920
510 FORJ=1TOTT*10
530 SYS7512
540 IFPEEK(7525)=254THEN590
550 NEXTJ
560 GOSUB900
570 GOTO460
590 FORRR=5TOM
600 TT=TT-1
610 IFTT<=0THEN810
620 POKES+22*12+RR,28
630 POKES+22*12+RR,0
640 IFRR=MTHENGOSUB930:GOSUB900:
    GOTO660
650 NEXTRR
660 POKE36878,15:POKE36877,215:FORT=
    1TO200:NEXT:POKE36877,0
670 POKE36878,10:POKE36877,215:
    FORT=1TO300:NEXT:POKE36877,0
690 SC=SC+(30-M):Y$=STR$(SC)
700 FORX=1TOLEN(Y$)

```



```

710 POKE7713+X,ASC(MID$(Y$,X,1)):
    POKE38433+X,6
720 NEXT
725 GOTO450
730 M=5:GOSUB920:IFTA<>2THENTA=2:
    HS=SC:HS$=N$:GOTO765
732 IFN$=""THEN765
735 IFSC>HSTHENHS=SC:HS$=N$
740 Y$="HIGH SCORE =" +STR$(HS)
745 FORX=1TOLEN(Y$):GOSUB950:POKE7756+
    X-INT(LEN(Y$)/2),A1
750 POKE7756+DI+X-INT(LEN(Y$)/2),6:NEXT
755 Y$="BY " +HS$:FORX=1TOLEN(Y$):
    GOSUB950:POKE7800+X-INT(LEN(Y$)/2),A1
760 POKE7800+DI+X-INT(LEN(Y$)/2),6:NEXT
765 Y$="ANOTHER GAME ?":FORX=1TOLEN(Y$):
    GOSUB950:POKE7837+X,A1:POKE7837+
    DI+X,6:NEXT
770 GETA$
775 IFA$="Y"THEN788
780 IFA$="N"THEN870
785 GOTO770
788 Y$+"NAME? (19 LETTERS MAX)":
    FORX=1TOLEN(Y$):GOSUB950:
    POKE7833+X,A1
789 POKE7833+X+DI,6:NEXT
790 X=0
791 GETA$:IFA$=""THEN791
792 IFA$=CHR$(13)THEN796

```

```

793 A1=ASC(A$):GOSUB955:POKE7879+
    X,A1:POKE7879+X+DI,6:X=X+1
794 IFX>18THENFORX=7879TO7899:
    POKEX,32:NEXT:GOTO790
795 X$=CHR$(A1):L$=L$+X$:GOTO791
796 FORX=7714TO7719:POKEX,32:NEXT:
    GOSUB900
797 IFL$<>""THENNN$=L$:L$=""
800 FORX=7748TO7899:POKEX,32:NEXT:
    GOTO140
810 POKES+22*12+RR,0:GOSUB900
820 FORJ=RRTO21
830 POKES+22*12+J,28
840 POKES+22*12+J,0
850 NEXTJ
860 GOTO460
870 END
900 POKES+22*12+M,0:POKES+22*13+M,0:
    RETURN
920 POKES+22*13+M,30:POKES+22*12+M,30:
    POKES+22*13+M,31:RETURN
930 POKES+22*12+M,27:POKES+22*13+M,29:
    FORX=0TO100:NEXT:RETURN
950 A1=ASC(MID$(Y$,X,1))
955 IFA1>63ANDA1<96THENA1=A1-64
960 RETURN

```

The Positron Blaster is controlled by pressing the space bar. You can only shoot after the creature appears.

Variables in this Program Include:

CL=Start of color code memory table
 CS=Start address of new character set
 DI=CL-S or number to add to S to get color code table
 S=Screen start address
 X\$=String constant "SCORE ="
 A1=Screen code number
 HS=Highest score
 HS\$=String of players name having highest score
 M=X coordinate of Slime Creature
 N\$=Present players name
 TT=Time allowed for player to fire

How the Program Works

- 2-3 Clears screen and prints instructions
- 5 Disables run/stop routine, allowing keyboard polling by machine language
- 10 Protects 512 bytes below the screen start address for new character set
- 20 Determines start address of new character set
- 30 Transfers first 65 characters of standard upper case from ROM to new character address
- 32-34 Loads machine-language routine for polling keyboard
- 40-116 Changes selected screen characters to characters needed for the game
- 120 Initializes custom character set
- 121 Lets player input name for keeping track of highest score
- 122 Prints green border for score

- 125 Prints "SCORE=" and then prints "(HOME)" to move cursor back to start
- 130-140 Puts constants into variable table and initializes variables. I is left screen, S is screen start address, CL is start of color code memory table, and DI is difference (CL-S).
- 145 TA=2 is key to bypass drawing screen after first game
- 290-370 Draws swamp
- 440 Prints Jacque on screen
- 450 Randomly selects Slime Creature's start position
- 460 Randomly determines how far Slime Creature moves each time
- 462-463 Timing loop that randomly determines how long the Slime Creature is under water
- 465 Determines if a Slime Creature got Jacque, if so, then to another game
- 470 To subroutine to erase Slime Creature
- 472-477 Creates a random-length series of tone bursts, in which the length is proportional to how long the Slime Creature is emerged
- 490 To subroutine to print a Slime Creature
- 500-550 TT gives random amount of time for player to fire. SYS at 530 is BASIC's entry command to a machine-language program.
- 540 peeks to see if space bar is pressed.
- NOTE: to make game more difficult, line 510 can be changed by decreasing TT's multiplier (10) to a smaller number
- 560 To subroutine to erase the Slime Creature
- 570 Goes back and moves the Slime Creature again

590-650 Moves Positron Blast, and if time runs out (TT=0), then the Slime Creature disappears and Blast continues to end of swamp. If not, then subroutine 930 prints an explosion, 900 erases it, and 660-670 is the sound of the explosion

690-720 Prints score

725 Goes back and repeats routine

730 TA=2 is key to by-pass drawing screen after first game. The Slime Creature is shown grabbing Jacque

731 By-passes routine to print high score and by whom if name has not been entered

735 HS is used to store high score, and HS\$ stores player's name

740-750 Prints "BY WHOM"

765 Asks about another game

770-785 Inputs player's decision

788-797 These lines use a method other than "input" to enter player's name. This is to prevent messing up playing screen if player enters a name longer than 19 letters, which causes printing on the next line in the scene area

788-789 Prints "NAME?" and the warning of 19 letters only

790 Makes sure X (used to count letters) starts at 0

791 Causes computer to wait for input

792 Ends the input if restore (CHR\$(13)) is pressed

793 Pokes in one letter and advances X (the counter)

794 Erases the input if the number of letters exceeds 19 and goes back to wait for new input

795 Keeps adding letters to the string L\$ for the name and goes back for the next until the player presses restore

796 Erases the name and goes back to 900 to the erase the Slime Creature routine

797 Sets string N\$ equal to L\$ and restores L\$ string to null or 0 ready for next input

800-810 Erases printing and returns for another game

820-850 Routine to move blast to end of swamp

860 Back to move a Slime Creature again

900 Erases Slime Creature subroutine

920 Get Slime Creature subroutine

930 Explosion subroutine

950-960 Converts CHR\$ code numbers to screen code numbers

How You Can Change the Program

1. Allowing multiple Slime Creatures to appear and disappear. You would allow Jacque to move on the screen to position him in the row of the creature you want to shoot.
2. Limiting the number of positron blasts and Slime Creatures per game. For example, allow 20 Slime Creatures and 20 Positron Blasts to see if Jacque can hit them all. You should also display the Positron Blasts remaining.